Particular solution to a Riccati equation $y' = 1 + 2y + xy^2$ The equation is $y' = 1 + 2y + xy^2$.  
I've tried $mx+n$, $ax^m$, even $\tan x$ as candidates for particular solution where $a,m,n \in \mathbb Q$, but it did not work. Can anyone find one particular solution?
It is not homework. Thank you in advance.
 A: This  link  solve it . The General Solutions of Linear ODE and Riccati Equation http://arxiv.org/abs/1006.4804 
A: Let us consider a generic Riccati equation:
\begin{equation}
y^{'}(x)= q_0(x) + q_1(x) y(x) + q_2(x) y(x)^2
\end{equation}
As Wikipedia teaches us the substitution $y(x)=-1/q_2(x) u^{'}(x)/u(x)$ reduces the ODE above to a following 2nd order linear ODE:
\begin{equation}
u^{''}(x) - u^{'}(x)\left(\frac{q_2^{'}(x)}{q_2(x)}+q_1(x) \right) + q_0(x) q_2(x) u(x) = 0
\end{equation}
Now the standard procedure is to reduce the linear ODE above to the normal form (i.e. such where the coeffiecient at the first derivative vanishes). This is done via $u(x)=\sqrt{q_2(x)} \exp(1/2 \int q_1(x) dx) \cdot v(x)$. This results in a following ODE:
\begin{equation}
v^{''}(x) + \frac{1}{4}\left( 4 q_0(x) q_2(x)+2 q_1'(x)-\frac{2 q_1(x) q_2'(x)}{q_2(x)}-q_1(x)^2+\frac{2 q_2''(x)}{q_2(x)}-\frac{3 q_2'(x)^2}{q_2(x)^2}\right) \cdot v(x)=0
\end{equation}
At this stage we have to input our coefficients. We consider a slightly more generic equation given by:
\begin{eqnarray}
q_2(x)&=&x\\
q_1(x)&=&2\\
q_1(x)&=& 1+a x
\end{eqnarray}
Then the coefficient at $v(x)$ in the ODE above reads:
\begin{equation}
\frac{1}{4} \left(\dots \right)= a x^2+x-1-\frac{1}{x} - \frac{3}{4} \frac{1}{x^2}
\end{equation}
Now from Hunt for exact solutions of second order ordinary differential equations with varying coefficients. we know that the equation above is mapped onto the biconfluent Heun equation through a following substitution:
\begin{equation}
v(x)= \exp\left( \frac{-\imath}{2 \sqrt{A}} x(A_0+A x)\right) \cdot x^{1/2\left(1+\sqrt{1-4 B} \right)} \cdot w(x)
\end{equation}
where $(A,A_0,A_1,A_2,B)=(a,1,-1,-1,-3/4)$. By doing this we obtain the following :
\begin{equation}
w^{''}(x)+w^{'}(x) \left( -\frac{\imath}{\sqrt{a}}+\frac{3}{x} - 2 \imath \sqrt{a} x\right) + w(x) \left( \frac{-1-\frac{3 \imath}{2 \sqrt{a}}}{x}-4 \imath \sqrt{a}-\frac{1}{4 a}-1\right)=0
\end{equation}
This is already almost the biconfluent equation https://dlmf.nist.gov/31.12 except for one subtelty, we have to rescale the variable $x$ appropriately . We take ${\mathfrak T}:= (-1)^{1/4}/(\sqrt{2} a^{1/4})$ and  we change the abscissa $x \rightarrow {\mathfrak T} x$ and $d/d x \rightarrow 1/{\mathfrak T} d/dx$ . This results in:
\begin{equation}
w^{''}(x)+w^{'}(x) \left( \frac{\frac{1}{2}-\frac{i}{2}}{a^{3/4}}+x+\frac{3}{x}\right) + w(x) \left( \frac{(3-3 i)-(2+2 i) \sqrt{a}}{4 a^{3/4} x}-\frac{i}{8 a^{3/2}}-\frac{i}{2 \sqrt{a}}+2\right)=0
\end{equation}
which is already the biconluent Heun equation.
For those who do not trust my calculations I enclose a Mathematica code snippet that reproduces those results.
Clear[y]; Clear[q2]; Clear[q1]; Clear[q0]; Clear[u];

y[x_] = -1/q2[x] u'[x]/u[x];
ll = Numerator[Together[y'[x] - (q0[x] + q1[x] y[x] + q2[x] y[x]^2)]];
myeqn = Collect[ll/Coefficient[ll, u''[x]], {u[x], u'[x], u''[x]}, 
  Simplify]



Clear[m]; Clear[u]; Clear[v]; Clear[w];
m[x_] = FullSimplify[
  Exp[-1/2 Integrate[
     Coefficient[myeqn, u'[x]]/Coefficient[myeqn, u''[x]], x]]]
u[x_] = m[x] v[x];
ll = Simplify[myeqn/m[x]];
myeqn1 = Collect[ll/Coefficient[ll, v''[x]], {v[x], v'[x], v''[x]}, 
  Simplify]

q2[x_] = x;
q1[x_] = 2;
q0[x_] = 1 + a x;
Collect[Coefficient[myeqn1, v[x]], x^_]

{A, A0, A1, A2, B} = {a, 1, -1, -1, -3/4};
M[x_] = Exp[-I/(2 Sqrt[A]) x (A0 + A x)] x^(1/2 (1 + Sqrt[1 - 4 B]));
v[x_] = M[x] w[x];
myeqn2 = Collect[Simplify[myeqn1/M[x]], {w[x], w'[x], w''[x], x^_}, 
  Expand]

Clear[f]; TT = (-1)^(1/4)/(Sqrt[2] a^(1/4));
f[x_] = TT x;
subst = {x :> f[x], 
   Derivative[1][w][x] :> 1/f'[x] Derivative[1][w][x], 
   Derivative[2][w][x] :> -f''[x]/(f'[x])^3 Derivative[1][w][x] + 
     1/(f'[x])^2 Derivative[2][w][x]};
Collect[TT^2 (myeqn2 /. subst /. w[f[x]] :> w[x]), {w[x], w'[x], 
  w''[x], x^_}, Simplify]


